IBM POUGHKEEPSIE

Diagnostic Engineering Publication

1410/7010

December 1, 1963

Subject: Diagnostic Program WT01B 1415 I/O Printer Test

Sequence Number 551
Replaces WT01A

When WT01 is in card form card # 001 is a System Control Card. It does not have any control information punched in it when it is released.

Refer to "1410/7010 Introduction", Volume 1.00 for instructions on how it must be punched.

This is a modified and improved version of WT01A. The modifications include:

A. Changes necessary to be compatible with the current diagnostic format.
B. Removal of the test routine called "WMS AND BLANKS IN M & L MODES."
C. Alteration and expansion of the test routine called "WM ALIGNMENT AND WM PERIOD TESTS."
D. Inclusion of a new test routine to check on band width (detenting difference) and alignment.
E. Changing the timing section to type out the time it took to type each line instead of each pair of lines. The timing routine (now) covers 7010 as well as 1410 systems.
F. Changing the method in which the optional "SELECTED CHARACTER ROUTINE" (build your own test pattern routine) operates.

Enclosures: 26 Pages
Card Deck for CARD ONLY SYSTEMS (as punched by UP51)
8 Cards - Card Loader (1-7) and 1Core Clear
62 Cards No. 001-062 Data Cards
1 Card Execute Card

Distribution: X 1410
X 7010
Other

WT01

1415 CONSOLE I/O PRINTER TEST

(1410/7010)

December 1, 1963

CONTENTS OF WT01 WRITE-UP AND LISTING

5.00.00.0 TEST DESCRIPTION

00.1 MODIFICATIONS

This is a modified and improved version of WT01A. The modifications include:

A. Changes necessary to be compatible with the current diagnostic format. (Standard TADs at location 01000 and a Standard System Control Card to provide necessary system information and eliminate unnecessary operator intervention.)

B. Removal of the test routine called "WMS AND BLANKS IN M & L MODES." This test routine contributed little to the overall effectiveness of the test.

C. Alteration and expansion of the test routine called "WM ALIGNMENT AND WM PERIOD TESTS." See description, Section 5.00.00.2, for further information.

D. Inclusion of a new test routine to check on band width (detenting difference) and alignment.

E. Changing the timing section to type out the time it took to type each line instead of each pair of lines. The timing routine (now) covers 7010 as well as 1410 systems.

F. Changing the method in which the optional "SELECTED CHARACTER ROUTINE" (build your own test pattern routine) operates. See OPERATING PROCEDURES, Section 5.00.02.2.

00.2 DESCRIPTION

WT01 is a functional test of the Program Printout Operations of the 1415 Console I/O Printer on the 1410 or 7010 Data Processing System. Test routines are directed toward checking Character Printout, Space, Word-Mark Control, and Carriage Return and Indexing Operations. The Input Operation is tested through the use of the Console Inquiry function.

5.00.00.0 TEST DESCRIPTION (continued)

Test patterns are designed to test specific operations or phases of operations. Before each pattern is typed, the title of the test pattern selection character is typed (see Section 5.00.02.2 for use of test pattern selection character).

The test patterns, their titles and test objectives are explained in the order in which they are run. Each test line of characters is typed twice for (visual) comparison.

COLLATING SEQUENCE A

Type all characters in the COLLATING SEQUENCE for convenient visual checking.

ROCK B

Test the tilt mechanism by typing the characters located one after the other in vertical columns on the print head.

ROLL C

Test the rotate mechanism by selecting characters one after the other in horizontal bands around the print head.

TWIST D

Test the combined rotate and tilt mechanism by causing a maximum rotation and tilt between characters.

WM ALLIGNMENT AND WM PERIOD TESTS E

Exercise thoroughly spacing and backspacing mechanisms by typing word marks over every other character and then over every character. The word-mark period latch is given specific attention here.

BANDWIDTH & ALIGNMENT TEST F

The characters typed are chosen specifically to test band width (detenting difference), alignment and the action of the wear compensator. The characters, $!QNLJ, are chosen because of their rotate selections. If a band width exists, it will be greatest among these characters. They are also used in a final check during alignment (fine tuning). The "J" is used extensively to cause the wear compensator to take up slack in the rotate and select system.

5.00.00.0 TEST DESCRIPTION (continued)

All test pattern selection characters should line up in position 42 on the margin scale as a test of the spacing operation.[1]

Carriage return is always tested in two ways, by margin lever stop and again by a group mark word mark at the end of the write field. All fixed test patterns are 83 characters long. Because of the printout identification character (R normally) and the space that follows it, the first test pattern character is typed in position three and the last in position eighty-five if the tabs are set correctly. A carriage return and indexing operation is therefore initiated by both the B channel group mark word mark and an end of line condition. This produces a double space between each pair of lines of every test pattern. Look for this to occur.

00.3 EQUIPMENT

Any model 1410 or 7010 Data Processing System. The 1415 Console I/O Printer is the only I/O device tested. It is assumed to be on E channel only.

The Processing Overlap Feature is not necessary but is done in overlap mode if it is available.

00.4 CARD DECK

A complete card deck of WT01 consists of the following:

7	cards	Loader
1	card	Execute (Core Clear)
	program cards[2]	Program WT01
1	card	Execute (branch to 02000)

Note: Card No. 001 is a System Control Card. It does not have any control information punched in it when it is released. See "1410/7010 Introduction," Volume 1.00, for instructions on how to punch it.

00.5 EC LEVEL OF MACHINE

Not applicable.

---

1. Be sure to follow instructions on setting up margin lever stops as explained in OPERATING PROCEDURES, Section 5.00.02.1.
2. See Release sheet for exact number of cards.

5.00.01.0 LOADING PROCEDURES

Use Standard Diagnostic Loading Procedure. Refer to "1410/7010 Introduction," Volume 1.00, for further information.

5.00.02.0 OPERATING PROCEDURES

02.1 Always set the right and left hand margin lever stops to their maximum right and left hand positions (0 and 85, respectively). The test patterns and the character position count both depend on this. A group of four-digit numbers separated by slashes occurs in one line of this test for counting purposes. The units position of each number corresponds to the position of the character with respect to the left-hand margin. The printout identification character R is counted as number one.

WT01 begins immediately on completion of loading and no manual intervention is required.

02.2 Test operation can be altered at any time by using the "Program Alter Routine." An Inquiry Request is acknowledged upon completion of any line of type. TADs are loaded as blanks and the locations are only tested for 1. TAD5, a Special TAD, is an exception and its use is described fully.

Standard TADs

TADs	Address	Not 1	1
TAD0	01000	Do Not	Bypass Typeouts
TAD1	01001	Do Not	Loop on Routine
TAD2	01002	Do Not	Halt on Error
TAD3	01003	Do Not	Repeat Test

Special TADs

TAD4	01004	Do Not	Typeout time to type 1 line
TAD5	01005	Do Not	Select Test Pattern by letter

TAD 0 is used only to bypass an error message typeout.

Setting TAD 4 to a 1 causes a typeout of the time it took to type the line preceding it to be given. Use only on systems with the Processing Overlap Feature.

5.00.02.0 OPERATING PROCEDURES (continued)

Use TAD 5 to select a particular test pattern by name (actually by letter). If it remains a blank, all test routines are run in order. Entering the test pattern selection character (A, B, C, ... F) causes the test to go directly to the pattern selected. The test patterns and the letters that relate to them are covered in the description, Section 5.00.00.1. Entering an X causes the test to go to the "SELECTED CHARACTER ROUTINE." After entering an M or an L in response to "ENTER MODE- M OR L," the request "ENTER DATA FIELD" is made. At this time a full line of characters with or without word marks may be entered. If the number of characters entered is less than a full line (83), the portion entered is expanded to produce a full line typeout. To have less than a full line typed out, enter a group mark word mark after the last character to be typed. The line of characters is typed twice unless TAD1 is set to loop on routine. Entering a Z in TAD5 takes the program to the end of job message and into the next test.

5.00.03.0 OPERATING HINTS, COMMENTS

03.1 On systems equipped with overlap all test routines are typed in overlap mode. This makes it convenient to give typeouts of the length of time it takes to type a given line on request. If for some reason it is necessary to operate in unoverlap mode once the test is in progress, alter location 01263 to a blank (location denotes overlap in System Control Card), RESET and START. The test is started over from the beginning including the necessary initialization.

Should it ever be necessary to time (approximately) a carriage return operation instead of a normal line print operation, the following is offered.[1] Use the SELECTED CHARACTER ROUTINE to type a simple line, preferably blanks (b's) in Load Mode or zeros (0's). Set TAD 1 to loop on routine (location 01001 to a 1) and TAD4 to a 1 for timing. With the right hand margin selector on 85 (end of line), take several lines of outputs. Now set the margin selector to 84. This causes a carriage return and the last character of the line to be typed in column 1. The time difference between the two lines is carriage return time (approximately).

---

1. Timing can only be used on systems with the Processing Overlap Feature.

5.00.03.0 OPERATING HINTS, COMMENTS

03.2 The time for one pass of WT01 including all test routines, titles, etc., but no timing typeouts or Inquiry Requests is approximately 4 minutes.

03.3 The SELECTED CHARACTER ROUTINE can be used to investigate the Output Error Routine by entering a group mark word mark for the data field. This causes an underscored zero (0) followed by underscored blanks (b) to be typed. All characters are typed in column 1. Once this operation is initiated, it is not under program control and STOP or RESET must be used to terminate it.

5.00.04.0 PROGRAM STOPS, RESTARTS

There are no Normal Stops in WT01 and only one Error Stop. It is under TAD control and occurs only if TAD 2 is set to 1. The STOP follows an error typeout indicating a data check error. Push START to continue the test.

RESET and START causes the test to begin again at 02000, repeating the typeout of the test identification and performing all the initialization.

5.00.05.0 TYPEOUTS

05.1 The only typeout that has not been explained in preceding sections or may need clarification is:

*** DATA CHECK IN LAST LINE TYPED ***

This message indicates that a parity check error (Data Check) occurred during the typing of the test line above it. The character or characters involved should be underscored.

APPENDIX

1415 CONSOLE PRINTER

TRANSLATOR, OUTPUT

BCD Bits	Magnet Picked
$\overline{2}$	R1
$\overline{8} \cdot 4$	R2
$\overline{8} + 4$	R2A
$8 \cdot \overline{1} + \overline{8} \cdot 1$	R5
$\overline{A}$	T1
$\overline{B}$	T2
$\overline{C}$	CK
$\overline{8 \cdot 4 \cdot 2 \cdot 1 + 8 \cdot 4}$	UC
All others	LC
v (Word Mark)	UC• CK
_ (Underscore)	UC• CK• T1• T2

TRANSLATOR, INPUT

Contacts Transferred	BCD Bit
$R5 \cdot \overline{R2A} \cdot LC + \overline{R5} \cdot R2A + \overline{R5} \cdot UC$	1
$R1 \cdot \overline{R2A} \cdot + LC \cdot R1$	2
$R2 \cdot \overline{R2A}$	4
$R2A \cdot LC + \overline{R2A} \cdot UC$	8
T1	A
T2	B
CK + Space	C
Word Mark	WM

Contracts transfer when corresponding magnet is NOT picked, except R5 which transfers when magnet is picked.
Keyboard to contact coding is same as magnets picked.

1415 CONSOLE PRINTER

Character	BCD Code	Magnets Picked
b (Blank)	C	R1 R2 R2A T1 T2 UC *
. (Period	B A 8 2 1	C LC
¤)	C B A 8 4	R1 R2A R5 UC
[	B A 8 4 1	R1 R2A C UC
<	B A 8 4 2	R2A R5 C UC
‡ (Group Mark)	C B A 8 4 2 1	R2A UC
& (Ampersand) +	C B A	R1 R2 R2A UC *
$	C B 8 2 1	T1 LC
*	B 8 4	R1 R2A R5 T1 C UC
]	C B 8 4 1	R1 R2A T1 UC
;	C B 8 4 2	R2A R5 T1 UC
Δ	B 8 4 2 1	R2A T1 C UC
-	B	R1 R2 R2A T1 C UC *
/	C A 1	R1 R2 R2A R5 T2 LC *
, (Comma)	C A 8 2 1	T2 LC
% ( ) (	A 8 4	R1 R2A R5 T2 C UC
~ (Wd Separator)	C A 8 4 1	R1 R2A T2 UC
\	C A 8 4 2	R2A R5 T2 UC
## Segment Mark	A 8 4 2 1	R2A T2 C UC
¢ Substitute	A	R1 R2 R2A T2 C UC *
# Blank =	8 2 1	T1 T2 C LC
@ ,	C 8 4	R1 R2A R5 T1 T2 UC
:	8 4 1	R1 R2A T1 T2 C UC
>	8 4 2	R2A R5 T1 T2 C UC
⌐ (Tape Mark)	C 8 4 2 1	R2A T1 T2 UC
?	C B A 8 2	R5 LC
A	B A 1	R1 R2 R2A R5 C LC
B	B A 2	R2 R2A C LC
C	C B A 2 1	R2 R2A R5 LC
D	B A 4	R1 R2A C LC
E	C B A 4 1	R1 R2A R5 LC
F	C B A 4 2	R2A LC
G	B A 4 2 1	R2A R5 C LC
H	B A 8	R1 R5 C LC
I	C B A 8 1	R1 LC
!	B 8 2	R5 T1 C LC
J	C B 1	R1 R2 R2A R5 T1 LC
K	C B 2	R2 R2A T1 LC
L	B 2 1	R2 R2A R5 T1 C LC
M	C B 4	R1 R2A T1 LC
N	B 4 1	R1 R2A R5 T1 C LC
O	B 4 2	R2A T1 C LC
P	C B 4 2 1	R2A R5 T1 LC

* From keyboard R5 selected instead of R1, R2, R2A.

1415 Console Printer (continued)

Character	BCD Code	Magnets Picked
Q	C B 8	R1 R5 T1 LC
R	B 8 1	R1 T1 C LC
≠ (Record Mark)	A 8 2	R5 T2 C LC
S	C A 2 1	R2 R2A R5 T2 C LC
T	A 2 1	R2 R2A R5 T2 C LC
U	C A 4	R1 R2A T2 LC
V	A 4 1	R1 R2A R5 T2 C LC
W	A 4 2	R2A T2 C LC
X	C A 4 2 1	R2A R5 T2 LC
Y	C A 8	R1 R5 T2 LC
Z	A 8 1	R1 T2 C LC
0	C 8 2	R5 T1 T2 LC
1	1	R1 R2 R2A R5 T1 T2 C LC
2	2	R2 R2A T1 T2 C LC
3	C 2 1	R2 R2A R5 T1 T2 LC
4	4	R1 R2A T1 T2 C LC
5	C 4 1	R1 R2A R5 T1 T2 LC
6	C 4 2	R2A T1 T2 LC
7	4 2 1	R2A R5 T1 T2 C LC
8	8	R1 R5 T1 T2 C LC
9	C 8 1	R1 T1 T2 LC
v (Word Mark)		C UC
_ (Underscore)		T1 T2 C UC

```
R WTØ1B
R
R
R COLLATING SEQUENCE                        A
R
R     .. ¤¤ [[ << ‡‡ && $$ ** ]] ;; ΔΔ -- // ,, %% ~~ \\ ** bb ## @@ :: >> √√ ?? /ØØ85
R     .. ¤¤ [[ << ‡‡ && $$ ** ]] ;; ΔΔ -- // ,, %% ~~ \\ ** bb ## @@ :: >> √√ ?? /ØØ85
R
R AA BB CC DD EE FF GG HH II !! JJ KK LL MM NN OO PP QQ RR ‡‡ SS TT UU VV WW XX YY ZZ
R AA BB CC DD EE FF GG HH II !! JJ KK LL MM NN OO PP QQ RR ‡‡ SS TT UU VV WW XX YY ZZ
R
R ØØ 11 22 33 44 55 66 77 88 99 35 /ØØ4Ø/ØØ45/ØØ5Ø/ØØ55/ØØ6Ø/ØØ65/ØØ7Ø/ØØ75/ØØ8Ø/ØØ85
R ØØ 11 22 33 44 55 66 77 88 99 35 /ØØ4Ø/ØØ45/ØØ5Ø/ØØ55/ØØ6Ø/ØØ65/ØØ7Ø/ØØ75/ØØ8Ø/ØØ85
R
R ROCK                                      B
R
R #,$.IRZ96WOFDMU42SKB?!‡Ø8YQHGPX75VNECLT31/JA√*Δ‡[]~:bb-&<;\>@%*¤#,$.IRZ96WOFDMU42SK
R #,$.IRZ96WOFDMU42SKB?!‡Ø8YQHGPX75VNECLT31/JA√*Δ‡[]~:bb-&<;\>@%*¤#,$.IRZ96WOFDMU42SK
R
R ROLL                                      C
R
R #9642Ø87531√:b>@%\b~*/TVXY‡SUWZ,$ROMK!QPNLJΔ]-;*¤<&[‡ACEGH?BDFI.$,#9642Ø87531√:b>@%
R #9642Ø87531√:b>@%\b~*/TVXY‡SUWZ,$ROMK!QPNLJΔ]-;*¤<&[‡ACEGH?BDFI.$,#9642Ø87531√:b>@%
R
R TWIST                                     D
R
R @E%N*V¤5<7;X\P>@b?b!-‡&Ø[4]U~M:D√F*OΔW‡6‡.¤5<7&Ø[4‡6F√D:?bG>E@E%N*V¤5<7\XA.A.A.A.A.
R @E%N*V¤5<7;X\P>@b?b!-‡&Ø[4]U~M:D√F*OΔW‡6‡.¤5<7&Ø[4‡6F√D:?bG>E@E%N*V¤5<7\XA.A.A.A.A.
R
R WM ALIGNMENT AND WM PERIOD TESTS          E
R
R ĬIĬIM̆MM̆M!̆!!̆!Y̆YY̆Y:̆::̆:W̆WW̆W‡̆‡‡̆‡X̆XX̆XΔ̆ΔΔ̆ΔV̆VV̆V.̆..̆.VV̆VV̆ΔΔ̆ΔΔ̆XX̆XX̆‡‡̆‡‡̆WW̆WW̆::̆::̆YY̆YY̆!!̆!!̆MM̆MM̆IĬIĬ
R ĬIĬIM̆MM̆M!̆!!̆!Y̆YY̆Y:̆::̆:W̆WW̆W‡̆‡‡̆‡X̆XX̆XΔ̆ΔΔ̆ΔV̆VV̆V.̆..̆.VV̆VV̆ΔΔ̆ΔΔ̆XX̆XX̆‡‡̆‡‡̆WW̆WW̆::̆::̆YY̆YY̆!!̆!!̆MM̆MM̆IĬIĬ
R
R V̆V̆V̆V̆Δ̆Δ̆Δ̆Δ̆.̆M̆M̆M̆M̆!̆!̆!̆!̆.̆V̆V̆V̆V̆Δ̆Δ̆Δ̆Δ̆.̆M̆M̆M̆M̆!̆!̆!̆!̆.̆.̆.̆.̆.̆.̆.̆.̆.̆.̆.̆.̆.̆.̆!̆!̆!̆!̆M̆M̆M̆M̆.̆Δ̆Δ̆Δ̆Δ̆V̆V̆V̆V̆.̆!̆!̆!̆!̆M̆M̆M̆M̆.̆Δ̆Δ̆Δ̆Δ̆V̆V̆V̆V̆
R V̆V̆V̆V̆Δ̆Δ̆Δ̆Δ̆.̆M̆M̆M̆M̆!̆!̆!̆!̆.̆V̆V̆V̆V̆Δ̆Δ̆Δ̆Δ̆.̆M̆M̆M̆M̆!̆!̆!̆!̆.̆.̆.̆.̆.̆.̆.̆.̆.̆.̆.̆.̆.̆.̆!̆!̆!̆!̆M̆M̆M̆M̆.̆Δ̆Δ̆Δ̆Δ̆V̆V̆V̆V̆.̆!̆!̆!̆!̆M̆M̆M̆M̆.̆Δ̆Δ̆Δ̆Δ̆V̆V̆V̆V̆
R
R BANDWIDTH & ALIGNMENT TEST                F
R
R JJLNQ!$JJJJJJJJJJLNQ!$JJJJJJJJJJLNQ!$JJJJJJJJJJLNQ!$JJJJJJJJJJLNQ!$JJJJJJJJJJLNQ!$J
R JJLNQ!$JJJJJJJJJJLNQ!$JJJJJJJJJJLNQ!$JJJJJJJJJJLNQ!$JJJJJJJJJJLNQ!$JJJJJJJJJJLNQ!$J
R                             *** END OF JOB ***
R
```

```
PGLIN     LABEL     OPCOD   OPERAND                                          CT  ADDRS  INSTRUCTION
1002      LOADER    EQU     400
1003
1004      *                 *****   STANDARD TADS     *****
1005                ORG     1000                                                 01000
1006      *                         NOT 1             1
1007      TAD0      DC      a a     DO NOT    BYPASS TYPE OUTS                1  01000
1008      TAD1              a a     DO NOT    LOOP ON ROUTINE                 1  01001
1009      TAD2              a a     DO NOT    HALT ON ERRORS                  1  01002
1010      TAD3              a a     DO NOT    REPEAT PROGRAM                  1  01003
1011
1012                        *TEST SET UP IN THE NOT 1 CONDITION*
1013                         AND WILL ONLY CHECK FOR A 1
1014
1015      *                 *****   SPECIAL  TADS     *****
1016
1017      TAD4      DC      a a     DO NOT    TYPEOUT TIME TO TYPE 1 LINE     1  01004
1018                                          USE ONLY IF SYSTEM HAS OVERLAP
1019      TAD5              a a     DO NOT    SELECT TEST PATTERN BY LETTER   1  01005
1020
1021                                          * THE FOLLOWING MAY BE USED IN
1022                                            TAD5 TO SELECT TEST PATTERNS
1023                                          A  TEST A  COLLATING SEQUENCE
1024                                          B  TEST B  ROCKING EXERCISE
1025                                          C  TEST C  ROLLING EXERCISE
1026                                          D  TEST D  TWISTING EXERCISE
1027                                          E  TEST E  WORDMARK ALIGNMENT
1028                                          F  TEST F  BANDWIDTH-ALIGNMENT
1029                                          X  TEST X  SELECTED CHARACTERS
1030                                          Z  THEEND  EOJ MESSAGE & B 400
1031      GMWM      DCW     aGMa                                             1  01006
```

```
PGLIN  LABEL    OPCOD  OPERAND          COMMENTS                                CT  ADDRS  INSTRUCTION
1033   *                *PROGRAM ALTER AND CONTROL ROUTINE
1034
1035   CONTRL   SBR    CTLXIT&5         STORE RETURN ADDR                        7  01007  G 01081 B
1036   ENTER    RCP    ADDRES&4         ENTER LOCATION TO BE ALTERED            10  01014  M %TO 01049 R
1037            BNT1   CTLXIT,T         INQ NOT FROM CONSOLE                     7  01024  R 01076 SBTMGM
1038            BEX1   ENTER,M          TRY AGAIN IF 1/2/4/8                     7  01031  R 01014 M
1039            BA1    ADDRES                                                    7  01038  R 01045 M
1040   ADDRES   RCPW   00000            ENTER DATA INTO ADDRES SPECIFIED        10  01045  L %TO 00000 R
1041            BEX1   ADDRES,S         ,M                                       7  01055  R 01045 SMGM
1042            BA1    *&1                                                       7  01062  R 01069 M
1043
1044            B      TSTSEL           CHECK ON ENTRY TO SELECT A TEST          7  01069  J 01083
1045
1046   CTLXIT   B      00000            RETURN TO PROGRAM                        7  01076  J 00000
1047   *
1048   *                               *****
1049   *
1050   TSTSEL   BCE    TESTA,TAD5,A     COLLATING SEQUENCE                      12  01083  B 02007 01005 A
1051            BCE    TESTB,TAD5,B     ROCK PATTERN                            12  01095  B 02160 01005 B
1052            BCE    TESTC,TAD5,C     ROLL PATTERN                            12  01107  B 02251 01005 C
1053            BCE    TESTD,TAD5,D     TWIST PATTERN                           12  01119  B 02342 01005 D
1054            BCE    TESTE,TAD5,E     WM ALIGNMENT & WM PERIOD TESTS          12  01131  B 02433 01005 E
1055            BCE    TESTF,TAD5,F     BANDWIDTH AND ALIGNMENT ROUTINE         12  01143  B 02555 01005 F
1056            BCE    TESTX,TAD5,X     SELECTED CHARACTER ROUTINE              12  01155  B 02653 01005 X
1057            BCE    THEEND,TAD5,Z    EOJ MESSAGE & B 400 - NEXT TEST         12  01167  B 02993 01005 Z
1058            B      CTLXIT           RETURN TO ALTER ROUTINE                  7  01179  J 01076
1059            H                       DEFINE PRECEDING BRANCH LENGTH           1  01186  .
```

```
PGLIN   LABEL    OPCOD  OPERAND                   CT  ADDRS  INSTRUCTION

1061             ORG    1230        CONTROL INFORMATION     01230
1062             DC     a         a                     15  01244
1063             DC     a55100a     SEQ# 551   5K   SYS1 ONLY   5  01249
1064    TESTID   DCW    aWT01a      *TEST IDENTIFICATION    4  01253
1065    LEVEL    DC     a8a,G                                1  01254
1066
1067             ORG    1256        *SYSTEM CONTROL CARD       01256
1068    SYS1     DC     a a         INDICATE SYSTEM TYPE     1  01256
1069                                0   1410 STD
1070                                I   1410 ACC
1071                                X   7010
1072                    a     a     NOT INTERROGATED         6  01262
1073                    a a         1-SYSTEM HAS OVERLAP     1  01263
1074                    a         a NOT INTERROGATED        15  01278
1075                    a       *a                          10  01288
1076             ORG    1289                                   01289
1077
1078    *               UTILITY TYPING AND SPACING ROUTINE
1079
1080    TYPEIT   SBR    TYPE&8      STORE ADDRESS OF MESSAGE  7  01289  G 01304 B
1081    TYPE     WCP    00000       TYPE MESSAGE             10  01296  M %T0 00000 W
1082             SBR    TYPEXT&5    STORE ADDRESS FOR RETURN  7  01306  G 01383 B
1083             BCB1   TYPE                                  7  01313  R 01296 2
1084             BA1    *&1         CONTINUE                  7  01320  R 01327 G M
1085             CW     SPACEX&1                              6  01327  □ 01358
1086    SPACE    SBR    SPACEX&6    EXIT WHEN SPACING         7  01333  G 01363 B
1087             WCP    ABLANK      ONE BLANK LOCATION       10  01340  M %T0 01385 W
1088             BA1    *-16                                  7  01350  R 01340 G M
1089    SPACEX   NOPWM                                        1  01357  N
1090             B      00000       EXIT WHEN SPACING         7  01358  J 00000
1091             SW     SPACEX&1                              6  01365  , 01358
1092             BNQ    CONTRL      TO CONTROL ROUTINE        7  01371  J 01007 Q
1093    TYPEXT   B      00000       EXIT WHEN TYPING SUBTITLES, ETC  7  01378  J 00000
1094
1095    ABLANK   DCW    a a,G       JUST FOR A SPACE          1  01385
```

```
PGLIN   LABEL    OPCOD  OPERAND                                                 CT  ADDRS  INSTRUCTION

1097    *               INITIALIZATION- DONE ON FIRST PASS ONLY
1098
1099    SETUP    CS     99                     CLEAR OUT TOP 100 ADDRESSES          6  01387  / 00099
1100             MRCWG  B2000,1                SET UP RESET RESTART BRANCH AT 1    12  01393  D 01612 00001 0
                                                                                                         L
1101             SW     95,25                  SET WMS IN INDEX REGISTERS          11  01405  , 00095 00025
1102             MLWB   95,90                  MOVE THEM ALL THE WAY THROUGH       12  01416  D 00095 00090 M
                                                                                              Q
1103             ZA     OTIME,TIME             U SEC/PASS IN TIMING LOOP,1410      11  01428  M 01703 03587
1104             BCE    CK4OLP,SYS1,0          SYSTEM IS STD 1410                  12  01439  B 01485 01256 0
                                                                                              Q
1105             ZA     ITIME,TIME             U SEC/PASS 1410 ACC                 11  01451  M 01707 03587
1106             BCE    CK4OLP,SYS1,I          SYSTEM IS 1410 ACC                  12  01462  B 01485 01256 I
                                                                                              Q
1107             ZA     XTIME,TIME             U SEC/PASS 7010                     11  01474  M 01711 03587
1108    CK4OLP   BCE    *&19,SYS1&7,           CHECK FOR OVERLAP                   12  01485  B 01515 01263
1109             SW     OVRLAP&1               SET UP FOR OVERLAP                   6  01497  , 03209
1110             MLCS   @a@,TYPETP&1           TYPE IN OVERLAP MODE                12  01503  D 04436 03199 3
1111             SW     PATRNX&84              SET ADDRESS                          6  01515  , 04436
1112             SAR    ENDOFX                 IN INDEX REGISTER                    7  01521  G 00049 A
1113             SW     TWTGP&40               SETTING WORDMARK  IN PATTERN         6  01528  , 04056
1114             SW     SPBSP1,SPBSP1&82       SET WMS IN TEST PATTERN             11  01534  , 04100 04182
1115             SW     SPBSP2,SPBSP2&82                                           11  01545  , 04184 04266
1116             MLWB   SPBSP1&82,SPBSP1&80  MOVE WMS OVER EVERY OTHER ONE         12  01556  D 04182 04180 M
1117             MLWB   SPBSP2&82,SPBSP2&81                                        12  01568  D 04266 04265 M
1118             MLCS   @$@,ENTERX&9           SET UP READ CONSOLE PRINTER         12  01580  D 04437 02797 3
1119             B      TYPEIT                                                      7  01592  J 01289
1120             DCW    @WT01B@,G                                                   5  01603
1121             B      TESTA                  BEGIN TEST PATTERN SEQUENCE          7  01605  J 02007
1122
1123     B2000   DCW    @J02000 @,G            RESET RESTART                        7  01612
1124             ORG    *&X00                                                          01700
1125    OTIME    DCW    &0167                  U SEC/PASS IN TIMING LOOP 1410       4  01703
1126    ITIME           &0133                  U SEC/PASS IN TIMING LOOP 1410I      4  01707
1127    XTIME           &0047                  U SEC/PASS IN TIMING LOOP 7010       4  01711
```

PGLIN	LABEL	OPCOD	OPERAND		CT	ADDRS	INSTRUCTION
1129		ORG	2000	PROGRAM STARTS HERE		02000	
1130	START	B	SETUP	INITIALIZATION-DONE 1ST PASS ONLY	7	02000	J 01387 .
1131							
1132							
1133	TESTA	B	SPACE	SPACING ROUTINE	7	02007	J 01333
1134		B	TYPEIT	COMMON UTILITY TYPING ROUTINE	7	02014	J 01289
1135		DCW	@COLLATING SEQUENCE	A@,G	40	02060	
1136							
1137	TYPEA	B	WCP	TYPE TEST PATTERN IN MOVE MODE	7	02062	J 03100
1138		DCW	CSGP1	COLLATING SEQUENCE GROUP 1	5	02073	03596
1139		B	WCP	TYPE TEST PATTERN IN MOVE MODE	7	02074	J 03100
1140		DCW	CSGP1		5	02085	03596
1141							
1142		B	SPACE		7	02086	J 01333
1143		B	WCP	TYPE TEST PATTERN IN MOVE MODE	7	02093	J 03100
1144		DCW	CSGP2	COLLATING SEQUENCE GROUP 2	5	02104	03680
1145		B	WCP	TYPE TEST PATTERN IN MOVE MODE	7	02105	J 03100
1146		DCW	CSGP2		5	02116	03680
1147							
1148		B	SPACE		7	02117	J 01333
1149		B	WCP	TYPE TEST PATTERN IN MOVE MODE	7	02124	J 03100
1150		DCW	CSGP3	COLLATING SEQUENCE GROUP 3	5	02135	03764
1151		B	WCP	TYPE TEST PATTERN IN MOVE MODE	7	02136	J 03100
1152		DCW	CSGP3		5	02147	03764
1153							
1154		BCE	TYPEA,TAD1,1	REPEAT PATTERN A	12	02148	B 02062 01001 1

```
PGLIN   LABEL   OPCOD   OPERAND          CT  ADDRS  INSTRUCTION
1156    TESTB   B       SPACE            7   02160  J 01333
1157            B       TYPEIT           7   02167  J 01289
1158            DCW     @RCCK            40  02213
1159
1160    TYPEB   B       WCPW             7   02215  J 03115
1161            DCW     ROKGP            5   02226  03848
1162            B       WCPW             7   02227  J 03115
1163            DCW     ROKGP            5   02238  03848
1164
1165            BCE     TYPEB,TAD1,1     12  02239  B 02215 01001 1
1166    *
1167    *
1168    *
1169    TESTC   B       SPACE            7   02251  J 01333
1170            B       TYPEIT           7   02258  J 01289
1171            DCW     @ROLL            40  02304
1172
1173    TYPEC   B       WCPW             7   02306  J 03115
1174            DCW     ROLGP            5   02317  03932
1175            B       WCPW             7   02318  J 03115
1176            DCW     ROLGP            5   02329  03932
1177
1178            BCE     TYPEC,TAD1,1     12  02330  B 02306 01001 1
1179    *
1180    *
1181    *
1182    TESTD   B       SPACE            7   02342  J 01333
1183            B       TYPEIT           7   02349  J 01289
1184            DCW     @TWIST           40  02395
1185
1186    TYPED   B       WCPW             7   02397  J 03115
1187            DCW     TWIGP            5   02408  04016
1188            B       WCPW             7   02409  J 03115
1189            DCW     TWIGP            5   02420  04016
1190
1191            BCE     TYPED,TAD1,1     12  02421  B 02397 01001 1
```

REPEAT PATTERN B

*****

TYPE TEST PATTERN IN LOAD MODE

ROCK GROUP

TYPE TEST PATTERN IN LOAD MODE

REPEAT PATTERN C

*****

TYPE TEST PATTERN IN LOAD MODE

ROLL GROUP

TYPE TEST PATTERN IN LOAD MODE

REPEAT PATTERN D

*****

TYPE TEST PATTERN IN LOAD MODE

TWIST GROUP

TYPE TEST PATTERN IN LOAD MODE

PGLIN	LABEL	OPCOD	OPERAND		CT	ADDRS	INSTRUCTION
1193	TESTE	B	SPACE	SPACING ROUTINE	7	02433	J 01333
1194		B	TYPEIT	COMMON UTILITY TYPING ROUTINE	7	02440	J 01289
1195		DCW	@WM ALIGNMENT AND WM PERIOD TESTS Ea,G		40	02486	
1196							
1197	TYPEE	B	WCPW	TYPE TEST PATTERN IN LOAD MODE	7	02488	J 03115
1198		DCW	SPBSP1	SPACE AND BACKSPACE GROUP 1	5	02499	04100
1199		B	WCPW	TYPE TEST PATTERN IN LOAD MODE	7	02500	J 03115
1200		DCW	SPBSP1	SPACE AND BACKSPACE GROUP 1	5	02511	04100
1201							
1202		B	SPACE		7	02512	J 01333
1203		B	WCPW	TYPE TEST PATTERN IN LOAD MODE	7	02519	J 03115
1204		DCW	SPBSP2	SPACE AND BACKSPACE GROUP 2	5	02530	04184
1205		B	WCPW	TYPE TEST PATTERN IN LOAD MODE	7	02531	J 03115
1206		DCW	SPBSP2	SPACE AND BACKSPACE GROUP 2	5	02542	04184
1207							
1208		BCE	TYPEE,TAD1,1	REPEAT PATTERN E	12	02543	B 02488 01001 1
1209	*						
1210	*			*****			
1211	*						
1212	TESTF	B	SPACE		7	02555	J 01333
1213		B	TYPEIT		7	02562	J 01289
1214		DCW	@BANDWIDTH & ALIGNMENT TEST Fa,G		40	02608	
1215							
1216	TYPEF	B	WCP		7	02610	J 03100
1217		DCW	BWAGP	BANDWIDTH AND ALIGNMENT GROUP	5	02621	04268
1218		B	WCP		7	02622	J 03100
1219		DCW	BWAGP	BANDWIDTH AND ALIGNMENT GROUP	5	02633	04268
1220							
1221		BCE	TYPEF,TAD1,1	REPEAT PATTERN F	12	02634	B 02610 01001 1
1222							
1223							
1224		B	THEEND	TEST X DONE ON REQUEST ONLY	7	02646	J 02993

```
PGLIN   LABEL    OPCOD  OPERAND                                          CT  ADDRS  INSTRUCTION
1226    TESTX    B      SPACE                                             7  02653  J 01333
1227             B      TYPEIT                                            7  02660  J 01289
1228             DCW    @SELECTED CHARACTER ROUTINE          x@,G        40  02706
1229
1230             S      BUMP1              ZERO INDEX REGS USED TO COUNT  6  02708  S 00069
1231             B      TYPEIT                                            7  02714  J 01289
1232             DCW    @ENTER MODE- M OR L@,G                           18  02738
1233             RCPW   MODE               ENTER MODE - M OR L           10  02740  L %T0 03419 R
1234             BEX1   *-16,M             TRY AGAIN ON 1/2/4/8/A        7  02750  R 02740 SMG
1235             BA1    *&1                                               7  02757  R 02764 M
1236             B      TYPEIT                                            7  02764  J 01289
1237             DCW    @ENTER DATA FIELD@,G                             16  02786
1238    ENTERX   RCPW   PATRNX             ENTER CHARACTERS FOR PATTERN  1C  02788  L %T0 04352 R
1239                                       ENTER GMWM FOR SHORT LINE
1240             SBR    NEXT1              STORE ADDR OF LAST CHAR ENTERED&1  7  02798  G 00059 BSMG
1241             BEX1   *-23,M             TRY AGAIN ON 1/2/4/8/A        7  02805  R 02788 M
1242             BA1    *&1                                               7  02812  R 02819
1243             C      NEXT1,&PATRNX      SEE IF ANY ENTRY WAS MADE     11  02819  C 00059 04442
1244             BE     TYPEX              NO TYPE OLD PATTERN            7  02830  J 02914 S
1245             S      &1,NEXT1           REDUCE B ADDR REG BY 1        11  02837  S 04443 00059
1246    CK4END   C      NEXT1,ENDOFX       CHECK FOR END OF PATTERN      11  02848  C 00059 00049
1247             BE     TYPEX              OK TYPE IT                     7  02859  J 02914 S
1248    EXPAND   MLCWS  PATRNX&BUMP1,0&NEXT1      EXPAND TO FULL LINE     12  02866  D 04LV2 00*M0 7 Q
1249             SBR    NEXT1              STORE ADDR OF LAST CHAR ENTERED&1  7  02878  G 00059 B
1250             A      &1,BUMP1           ADD TO COUNTERS               11  02885  A 04443 00069
1251             A      &2,NEXT1           STEP TO NEXT LOCATION         11  02896  A 04444 00059
1252             B      CK4END             SEE IF ITS ALL DONE            7  02907  J 02848
```

```
PGLIN   LABEL   OPCOD   OPERAND          CT  ADDRS  INSTRUCTION
1254    TYPEX   BCE     LMODE,MODE,L     12  02914  B 02957 03419 L
1255            B       WCP              7   02926  J 03100
1256            DCW     PATRNX           5   02937  04352
1257            B       WCP              7   02938  J 03100
1258            DCW     PATRNX           5   02949  04352
1259            B       *&25             7   02950  J 02981
1260
1261    LMODE   B       WCPW             7   02957  J 03115
1262            DCW     PATRNX           5   02968  04352
1263            B       WCPW             7   02969  J 03115
1264            DCW     PATRNX           5   02980  04352
1265
1266            BCE     TYPEX,TAD1,1     12  02981  B 02914 01001 1
1267    *
1268    *                                *****
1269    *
1270    THEEND  B       TYPEIT           7   02993  J 01289
1271            DCW     a                48  03047
1272            BNQ     CONTRL           7   03049  J 01007 Q
1273            BCE     TESTA,TAD3,1     12  03056  B 02007 01003 1
1274            B       LOADER           7   03068  J 00400
1275            H                        1   03075  .
1276    *                                *****
1277            ORG     *&X00                03100
```

```
PGLIN  LABEL   OPCOD  OPERAND                                                CT  ADDRS  INSTRUCTION
1279   *               TEST PATTERN TYPING ROUTINE
1280
1281   WCP     SBR    DATA          STORE ADDRESS OF DATA PATTERN            7  03100  G 00039 B
1282           B      SETOP         SET UP TYPE INSTRUCTION MODE             7  03107  J 03130
1283           DCW    @M@           MOVE MODE                                1  03114
1284
1285   WCPW    SBR    DATA          STORE ADDRESS OF TEST PATTERN            7  03115  G 00039 B
1286           B      SETOP         SET MODE OF TYPE INSTRUCTION             7  03122  J 03130
1287           DCW    @L@                                                    1  03129
1288
1289   SETOP   SBR    *&6           STORE M OR L OP CODE                     7  03130  G 03142 B
1290           MLCWS  0,TYPETP      SET MODE IN TYPE INSTRUCTION            12  03137  D 00000 03198 7
1291           CW     6&DATA        SET ADDRESS                              6  03149  ¤ 000M6
1292           SAR    RETURN        FOR RETURN TO TEST ROUTINE               7  03155  G 00029 A
1293           S      TOTAL         ZERO TIMING COUNTER                      6  03162  S 03595
1294           CS     BUFFER&82     CLEAR CUT OUTPUT AREA                    6  03168  / 03582
1295           MLNA   4&DATA,*&6    SET ADDRESS OF TEST PATTERN             12  03174  D 000M4 03191 /
1296           MRCWG  0,BUFFER      SET TEST PATTERN INTO OUTPUT AREA        12  03186  D 00000 03500 L
1297   TYPETP  WCPW   BUFFER        TYPE TEST PATTERN                       10  03198  L %TO 03500 W
1298   OVRLAP  NOPWM                                                         1  03208  N
1299           BOL1   TIMER                                                  7  03209  J 03230 1
1300           BCB1   TYPETP                                                 7  03216  R 03198 2
1301           B      CK4ERR                                                 7  03223  J 03248
1302   TIMER   A      TIME,TOTAL    ADD LOOP TIME TO TOTAL                  11  03230  A 03587 03595
1303           BOL1   *-17          RETURN WHILE OVERLAP IN PROCESS          7  03241  J 03230 1
1304   CK4ERR  BA1    ERRORT        BRANCH TO ERROR ROUTINE                  7  03248  R 03328 M
1305           BCE    EDITIT,TA04,1 EDIT TIME FOR TYPEOUT                   12  03255  B 03274 01004 1
1306           B      CK4INQ        NO TIME TYPEOUT                          7  03267  J 03314
1307   EDITIT  MLCWA  CTLFLD,RESULT&4 PREPARE RESULT FIELD                   12  03274  D 03425 03430 X
1308           MCE    TOTAL-4,RESULT&4 EDIT TOTAL FOR TYPING                 11  03286  E 03591 03430
1309           WCP    RESULT        TOTAL TIME FOR ONE LINE                 10  03297  M %TO 03426 W
1310           BA1    *-16                                                   7  03307  R 03297 M
1311   CK4INQ  BNQ    CONTRL        TO CONTROL ROUTINE                       7  03314  J 01007 Q
1312           B      O&RETURN      RETURN TO TEST RCUTINE                   7  03321  J 000#0
```

PGLIN	LABEL	OPCOD	OPERAND		CT	ADDRS	INSTRUCTION
1314	*		ERROR ROUTINE				
1315							
1316	ERRORT	BCE	CK4HLT,TAD0,1	BYPASS ERROR TYPEOUT	12	03328	B 03392 01000 1
1317		B	TYPEIT		7	03340	J 01289
1318		DCW	@*** DATA CHECK IN	LAST LINE TYPED ***@,G	37	03383	
1319		BNQ	CONTRL		7	03385	J 01007 Q
1320	CK4HLT	BCE	HALT,TAD2,1	HALT ON ERROR	12	03392	B 03411 01002 1
1321		B	*&2		7	03404	J 03412
1322	HALT	H			1	03411	*
1323		B	CK4INQ	RETURN TO TEST PATTERN TYPING	7	03412	J 03314
1324							
1325	*			CONSTANTS, OUTPUT AREA			
1326							
1327	MODE	DCW	@ @,G	MODE-M OR L	1	03419	
1328	CTLFLD		@ . 0@	EDIT CONTROL FIELD	5	03425	
1329	RESULT		@ . SECS@,G	TIME TO TYPE 1 LINE OF TEST GROUP	10	03426	
1330							
1331		ORG	*&X00	UP TO NEXT HIGHER CENTURY ADDRESS		03500	
1332	BUFFER	DA	1X83,G	TYPE AREA		03500	
1333	TIME	DCW	&0C00	MICROSECONDS PER PASS IN ADD LOOP	4	03587	
1334	TOTAL		@000000000@	TOTAL TIME	8	03595	

```
PGLIN   LABEL    OPCOD  OPERAND                                                CT  ADDRS  INSTRUCTION

1336    *               TEST PATTERNS
1337
                              LL LL GG                RR .. DD             WW
1338    CSGP1    DC     a   .. ¤¤ BB TT MM && $$ ** BB ;; LL -- // ., %% SSa   50  03596
                          BB SS SS          .. GG TT QQ
1339                    a SS MM BB ## aa .. TT MM MM /0085a,G                  33  03678
1340    CSGP2           aAA BB CC DD EE FF GG HH II :: JJ KK LL MM NN OO PPa   50  03680
1341                    a CQ RR ** SS TT UU VV WW XX YY ZZa,G                  33  03762
1342    CSGP3           a00 11 22 33 44 55 66 77 88 99 35 /0040/0045/0050/0a   50  03764
1343                    a055/0060/0065/0070/0075/0080/0085a,G                  33  03846
                                              Q,                   TSDGLR
1344    ROKGP           a#,$.IRZ96WOFDMU42SKBM.*08YQHGPX75VNECLT31/JAMMLMBBa   50  03848
                         W. S  L.BG
1345                    aS. B-&T;STa%*¤#,$.IRZ96WCFDMU42SKa,G                  33  03930
                                    T. G  BSWS         ,         DR .  L
1346    ROLGP           a#9642087531M. Ta%SBSM/TVXY*SUWZ,$RCMK.QPNLJLB-;*¤Ta   50  03932
                          LG        Q               T. G
1347                    a&BMACEGHMBDFI.$,#9642087531M. Ta%a,G                  33  04014
                                L . B G  QS,     L R W . T S D G     L   L
1348    TWTGP           aaE%N*V¤5T7,XSPTG MB.-*&OB4BUSM.DMFMOLWM6#.¤5T7&OB4a   50  04016
                         G  T .Q  G           L B
1349                    aM6FMD.M GTEaE%N*V¤5T7SXA.A.A.A.A.a,G                  33  04098
                                ''''    ....             DDDD           DDD
1350    SPBSP1          aIIIIMMMM....YYYY....WWWW****XXXXLLLLVVVV...VVVVLLLa   50  04100
                         D           ....       ''''
1351                    aLXXXX****WWWW....YYYY....MMMMIIIIa,G                  33  04182
                                DDDD        ''''     DDDD         ''''            ''
1352    SPBSP2          aVVVVLLLL.MMMM.....VVVVLLLL.MMMM.....................a   50  04184
                         ''        DDDD       ''''           DDDD
1353                    a..MMMM.LLLLVVVV.....MMMM.LLLLVVVVa,G                  33  04266
1354    BWAGP           aJJLNQ.$JJJJJJJJJJLNQ.$JJJJJJJJJJLNQ.$JJJJJJJJJJLNQa   50  04268
1355                    a.$JJJJJJJJJJLNQ.$JJJJJJJJJJLNQ.$Ja,G                  33  04350
1356    PATRNX          a                                                a     50  04352
1357                    a                              a,G                     33  04434
1358
1359    RETURN   EQU    1,X              ADDR OF RETURN TO TEST ROUTINE
1360    DATA     EQU    3,X              ADDR OF DATA FIELD TO BE TYPED
1361    ENDOFX   EQU    5,X              ADDR OF END OF TEST X PATTERN
1362    NEXT1    EQU    7,X              ADDR REG USED IN TEST X EXPANSION
1363    BUMP1    EQU    9,X              COUNT TO EXPAND PATTERN IN TEST X
1364
1365             END    START                                                              J02000
1365                    aaa                                                     1  04436
1365                    a$a                                                     1  04437
1365                    PATRNX                                                  5  04442  04352
1365                    &1                                                      1  04443
1365                    &2                                                      1  04444
                                      END OF ASSEMBLY
```

SUMMARY

SET UP

Set right and left hand margin selector tabs to their maximum positions, 0 and 85 on the margin scale, respectively.

LOADING

Use standard 1410/7010 Diagnostic Loading Procedure. Refer to "1410/7010 Introduction," Volume 1.00, for additional information.

CONTROL

The following Standard and Special TADs are available for program control. None need be set to run this test.

TADs	Address	Not 1	1
TAD 0	01000	Do Not	Bypass Typeouts
TAD 1	01001	Do Not	Loop on Routine
TAD 2	01002	Do Not	Halt on Error
TAD 3	01003	Do Not	Repeat Test
TAD 4	01004	Do Not	Typeout time to type 1 line (use only if system has overlap)
TAD 5	01005	Do Not	Select Test Pattern by letter

The following may be used in TAD 5 to select test patterns:

A Test A COLLATING SEQUENCE
B Test B ROCKING EXERCISE
C Test C ROLLING EXERCISE
D Test D TWISTING EXERCISE
E Test E WORDMARK ALIGNMENT
F Test F BANDWIDTH-ALIGNMENT
X Test X SELECTED CHARACTERS
Z THE END EOJ MESSAGE & B 400

SUCCESS INDICATIONS

No error typeout, test patterns A through F typed-all pass visual inspection, and the end of job message.

ERROR INDICATIONS

Only one error typeout is given:

*** DATA CHECK IN LAST LINE TYPED ***

All other error indications are in the form of incorrectly typed test patterns, character alignment and positioning, etc., and can only be found through careful visual inspection of the typed page(s).